

## 附录一：短语结构规则的形式描述语言

## ★ 形式语言的 BNF

<规则> ::= && <规则名> <产生式> [:: <约束>]  
 <规则名> ::= “{” <标识符> ””  
 <产生式> ::= <短语类标记> → <短语结构>  
 <短语结构> ::= <结构成分> { <结构成分> }  
 <结构成分> ::= [ <中心成分标识> ] <语类标记>  
                   | [ <中心成分标识> ] <语类标记> “<<终结符>>”  
 <中心成分标识> ::= !  
 <语类标记> ::= <词类标记> | <短语类标记>  
 <词类标记> ::= n|v|a|t|s|……  
 <短语类标记> ::= np|vp|ap|tp|……  
 <终结符> ::= <汉语词语>  
                   | <标点符号>  
                   | <整数>  
 <约束> ::= <约束式> {, <约束式> }  
 <约束式> ::= <合一等式>  
                   | <条件测试句>  
 <条件测试句> ::= IF <合一等式> {, <合一等式> } TRUE  
                   | IF <合一等式> {, <合一等式> } FALSE  
                   | IF <合一等式> {, <合一等式> } THEN <合一等式> {, <合一等式> } ENDIF  
                   | IF <合一等式> {, <合一等式> } ELSE <合一等式> {, <合一等式> } ENDIF  
                   | IF <合一等式> {, <合一等式> } THEN <合一等式> {, <合一等式> } ELSE  
   <合一等式> {, <合一等式> } ENDIF  
 <合一等式> ::= <内部变量>=<原子>  
                   | <内部变量>=<内部变量>  
                   | <外部变量>=<外部变量>  
                   | <外部变量>=<True>  
                   | <外部变量>=<False>  
                   | <外部变量>=<None>  
                   | <外部变量>=<外部变量> <属性项目> { <属性项目> }  
                   | <外部变量>!<外部变量> <属性项目> { <属性项目> }  
 <外部变量> ::= <根结点标记>  
                   | <位置标记><语类标记>  
                   | <外部变量>.<外部属性>  
 <根结点标记> ::= \$  
 <位置标记> ::= % | %% | %%% | ……  
 <内部变量> ::= <外部变量>.<内部属性>  
 <外部属性> ::= <语义关系范畴> | <句法关系范畴>  
 <内部属性> ::= <词语语法范畴> | <短语语法范畴> | <语义性质范畴>

<语义关系范畴> ::= 主体|客体|与事|工具|处所|……  
 <句法关系范畴> ::= 主语|谓语|述语|宾语|补语|……  
 <词语语法范畴> ::= 前名|后名|个体量词|单作谓语|形状语|ccat|……  
 <短语语法范畴> ::= zhuyu|binyu|zhxyu1|zhxyu2|被动|否定|语气|cpcat|……  
 <语义性质范畴> ::= 语义类|配价数|主体变化|客体变化|……  
 <属性项目> ::= @<外部属性> | @<内部属性>  
 <原子> ::= <布尔型原子> | <符号型原子> | <数字型原子> | <层级型原子>  
 <布尔型原子> ::= <True> | <False>  
 <符号型原子> ::= <Symbol\_OR> | <Symbol\_NOT>  
 <Symbol\_OR> ::= <Symbol\_symbol>{<OR><Symbol\_symbol>}  
 <Symbol\_NOT> ::= ~<Symbol\_symbol>{~<Symbol\_symbol>}  
 <数字型原子> ::= <Number\_OR> | <Number\_NOT>  
 <Number\_OR> ::= <Number\_number>{<OR><Number\_number>}  
 <Number\_NOT> ::= ~<Number\_number>{~<Number\_number>}  
 <层级型原子> ::= <Hierarchy\_OR> | <Hierarchy\_AND> | <Hierarchy\_NOT>  
 <Hierarchy\_OR> ::= <Hierarchy\_value>{<OR><Hierarchy\_value>}  
 <Hierarchy\_AND> ::= <Hierarchy\_value><Hierarchy\_value>  
 <Hierarchy\_NOT> ::= <Hierarchy\_value>~<Hierarchy\_AND>  
 | ~<Hierarchy\_value>~<Hierarchy\_AND>  
 <OR> ::= “|”  
 <True> ::= 是 | Yes | YES  
 <False> ::= 否 | No | NO  
 <None> ::= 无 | NONE  
 <Symbol\_symbol> ::= <标识符>  
 <Number\_number> ::= <整数>  
 <Hierarchy\_value> ::= <标识符>  
 <标识符> ::= <文字> { <文字>|<阿拉伯数字> }  
 <文字> ::= <拉丁字母> | <汉字>  
 <整数> ::= <阿拉伯数字> { <阿拉伯数字> }  
 <拉丁字母> ::= a|b|c|…|z|A|B|C|…|Z  
 <阿拉伯数字> ::= 0|1|2|…|9

★ 补充说明：（一）元语言符号及其意义

- (1) ::= 表示“定义为”。
- (2) [X] 表示“X是可选的”。
- (3) {X} 表示“X可重复0到n次”，n是正整数。
- (4) <X> 表示X是非终结符（被定义项或有待定义项）。
- (5) “X”表示其中的X不是元语言中所用的符号，而是普通终结符。
- (6) | 表示并列“或”关系。

（二）短语结构规则中所用符号及其意义

- (1) “&&”、“，”等是分隔符，“&&”表示规则开始，“，”用于分隔约束式。
- (2) ~ 表示逻辑“非”。
- (3) /\* …… \*/ 表示注释，一般作为备注参考，可以出现在规则任何位置。
- (4) “.” 相当于自然语言中的结构助词“的”。